*************************************************************************************;
*************************************************************************************;
* File-Name:      surveyanalysis.do            	    		          				*;
* Date:           31/07/2023                                      					*;
* Authors:        Quinn, Sattler & Weymouth                       					*;
* Purpose:        Replication of "Do Exchange Rates Influence Voting?"  			*;
* 				  Section 2: Individual-level analysis								*;
* Input Files:    surveydata.dta													*;
* Machine:        Mac OS 12.6    	                                    			*;
* Program: 		  Stata/SE 16.1     												*;
* 				  Required Stata Packages: coefplot, grc1leg, blindschemes          *;
*************************************************************************************;
*************************************************************************************;

clear
use "surveydata.dta"

* Figure 2
* --------
*
* - Vote intentions

reg e_vi ib0.xrtreat if country == "MEX"
reg e_vi ib0.xrtreat if country == "IND"
reg e_vi ib0.xrtreat if country == "AUS"
reg e_vi ib0.xrtreat if country == "USA"
reg e_vi ib0.xrtreat if country == "JPN"

* - National economy

reg e_ne ib0.xrtreat if country == "MEX"
reg e_ne ib0.xrtreat if country == "IND"
reg e_ne ib0.xrtreat if country == "AUS"
reg e_ne ib0.xrtreat if country == "USA"
reg e_ne ib0.xrtreat if country == "JPN"

* - Personal situation

reg e_pe ib0.xrtreat if country == "MEX"
reg e_pe ib0.xrtreat if country == "IND"
reg e_pe ib0.xrtreat if country == "AUS"
reg e_pe ib0.xrtreat if country == "USA"
reg e_pe ib0.xrtreat if country == "JPN"

* - Plot results
* 
* Note: we use the gsem command to be able to plot all countries in one figure using coefplot; 
*       results are identical to the individual regressions above

foreach var in e_vi e_ne e_pe {

	g `var'_AUS = `var' if country == "AUS"
	g `var'_IND = `var' if country == "IND"	
	g `var'_JPN = `var' if country == "JPN"
	g `var'_MEX = `var' if country == "MEX"
	g `var'_USA = `var' if country == "USA"

	gsem (`var'_MEX <- ib0.xrtreat) ///
		 (`var'_IND <- ib0.xrtreat) ///
		 (`var'_AUS <- ib0.xrtreat) ///
		 (`var'_USA <- ib0.xrtreat) ///
		 (`var'_JPN <- ib0.xrtreat)	 
		 
	estimates store `var'	
	
}	
	
coefplot e_vi, msymbol(O) omitted baselevel keep(*:) drop(_cons var*) ///
	coeflabels(0.xrtreat="No change" 1.xrtreat="Depreciation" 2.xrtreat="Appreciation", labsize(3)) ///
	eqlabels("{bf:Mexico}" "{bf:India}" "{bf:Australia}" "{bf:United States}" "{bf:Japan}", labsize(3.5) asheadings)	///
	mlabel mlabpos(11) mlabgap(*1.5) format(%9.2f) ///
	xline(0, lcolor(black)) xscale(range(-0.3 0.3)) xlabel(-0.3(0.1)0.3, labsize(3) format(%9.1f)) ///
	title("{bf: Vote intentions}", size(3.5)) ///
	xsize(5) ///
	scheme(plotplain) saving(e_vi_plot.gph, replace) 
		
coefplot e_ne, msymbol(O) omitted baselevel keep(*:) drop(_cons var*) ///
	coeflabels(0.xrtreat="No change" 1.xrtreat="Depreciation" 2.xrtreat="Appreciation", labsize(3)) ///
	eqlabels("{bf:Mexico}" "{bf:India}" "{bf:Australia}" "{bf:United States}" "{bf:Japan}", labsize(3.5) asheadings)	///
	mlabel(@b) mlabpos(11) mlabgap(*1.5) format(%9.2f) ///
	xline(0, lcolor(black)) xscale(range(-0.6 0.6)) xlabel(-0.6(0.2)0.6, labsize(3) format(%9.1f)) ///
	title("{bf: National economy}", size(3.5)) ///
	xsize(5) ///
	scheme(plotplain) saving(e_ne_plot.gph, replace) 	
		
coefplot e_pe, msymbol(O) omitted baselevel keep(*:) drop(_cons var*) ///
	coeflabels(0.xrtreat="No change" 1.xrtreat="Depreciation" 2.xrtreat="Appreciation", labsize(3)) ///
	eqlabels("{bf:Mexico}" "{bf:India}" "{bf:Australia}" "{bf:United States}" "{bf:Japan}", labsize(3.5) asheadings)	///
	mlabel(@b) mlabpos(11) mlabgap(*1.5) format(%9.2f) ///
	xline(0, lcolor(black)) xscale(range(-0.4 0.6)) xlabel(-0.4(0.2)0.6, labsize(3) format(%9.1f)) ///
	title("{bf: Personal situation}", size(3.5)) ///
	xsize(5) ///
	scheme(plotplain) saving(e_pe_plot.gph, replace) 

graph combine e_vi_plot.gph e_ne_plot.gph e_pe_plot.gph, ///
	row(1) imargin(medlarge)  ///
	graphregion(fcolor(white)) ///
	title("")	
graph display, xsize(9.25) scale(1.2) 	
graph save Figure2.gph, replace
	
erase e_vi_plot.gph 
erase e_ne_plot.gph 
erase e_pe_plot.gph

drop e_vi_* e_ne_* e_pe_*
drop _est_e_vi _est_e_ne _est_e_pe


* Figure A20
* ----------
*
* - Recode outcome variables (5 cat. -> 3 cat.)

g e_pm_simple = 1 if e_pm == 1 | e_pm == 2
replace e_pm_simple = 2 if e_pm == 3
replace e_pm_simple = 3 if e_pm == 4 | e_pm == 5

g e_cb_simple = 1 if e_cb == 1 | e_cb == 2
replace e_cb_simple = 2 if e_cb == 3
replace e_cb_simple = 3 if e_cb == 4 | e_cb == 5

g e_vi_simple = 1 if e_vi == 1
replace e_vi_simple = 2 if e_vi == 2
replace e_vi_simple = 3 if e_vi == 3

g e_ne_simple = 1 if e_ne == 1 | e_ne == 2
replace e_ne_simple = 2 if e_ne == 3
replace e_ne_simple = 3 if e_ne == 4 | e_ne == 5

g e_pe_simple = 1 if e_pe == 1 | e_pe == 2
replace e_pe_simple = 2 if e_pe == 3
replace e_pe_simple = 3 if e_pe == 4 | e_pe == 5

* - Plot

foreach country in AUS IND JPN MEX USA {

	preserve

	keep if country == "`country'"
	di "`country'"
	
	if "`country'" == "AUS" {
		local name = "Australia"
	}
	if "`country'" == "IND" {
		local name = "India"
	}
	if "`country'" == "JPN" {
		local name = "Japan"
	}
	if "`country'" == "MEX" {
		local name = "Mexico"
	}
	if "`country'" == "USA" {
		local name = "United States"
	}

	di "`name'"
	
	* - Government performance
	
	proportion e_pm_simple if xrtreat==1 
	estimates store dep
	proportion e_pm_simple if xrtreat==0
	estimates store stab
	proportion e_pm_simple if xrtreat==2
	estimates store app

	coefplot (dep, color(eltblue)) (stab, color(emidblue)) (app, color(edkblue)),  /// 
		vertical recast(bar) barwidth(0.25) finten(60) ///
		title("`name'", size(5)) ///
		yscale(range(0 0.6)) ylabel(0(0.1)0.6,labsize(4)) ytitle(Proportion, size(4)) ///
		xlabel(1 "Bad" 2 "Neither/Nor" 3 "Good", angle(45) labsize(4)) /// 
		citop citype(logit) ciopt(lcolor(black) lwidth(vthin)) ///
		legend(label(1 "Depreciation") label(3 "No change") label(5 "Appreciation") size(4))
	graph save "govperf_`country'", replace
	
	* - Central bank performance
	
	proportion e_cb_simple if xrtreat==1 
	estimates store dep
	proportion e_cb_simple if xrtreat==0
	estimates store stab
	proportion e_cb_simple if xrtreat==2 
	estimates store app

	coefplot (dep, color(eltblue)) (stab, color(emidblue)) (app, color(edkblue)),  /// 
		vertical recast(bar) barwidth(0.25) finten(60) ///
		title("`name'", size(5)) ///
		yscale(range(0 0.6)) ylabel(0(0.1)0.6,labsize(4)) ytitle(Proportion, size(4)) ///
		xlabel(1 "Bad" 2 "Neither/Nor" 3 "Good", angle(45) labsize(4)) /// 
		citop citype(logit) ciopt(lcolor(black) lwidth(vthin)) ///
		legend(label(1 "Depreciation") label(3 "No change") label(5 "Appreciation") size(4))
	graph save "cbperf_`country'", replace
	
	* - Vote intentions
	
	proportion e_vi_simple if xrtreat==1 
	estimates store dep
	proportion e_vi_simple if xrtreat==0 
	estimates store stab
	proportion e_vi_simple if xrtreat==2 
	estimates store app

	coefplot (dep, color(eltblue)) (stab, color(emidblue)) (app, color(edkblue)),  /// 
		vertical recast(bar) barwidth(0.25) finten(60) ///
		title("`name'", size(5)) ///
		yscale(range(0 0.75)) ylabel(0(0.1)0.7,labsize(4)) ytitle(Proportion, size(4)) ///
		xlabel(1 "Less likely" 2 "Same" 3 "More likely", angle(45) labsize(4)) /// 
		citop citype(logit) ciopt(lcolor(black) lwidth(vthin)) ///
		legend(label(1 "Depreciation") label(3 "No change") label(5 "Appreciation") size(4)) 
	graph save "vote_`country'", replace
	
	* - National economy
	
	proportion e_ne_simple if xrtreat==1 
	estimates store dep
	proportion e_ne_simple if xrtreat==0 
	estimates store stab
	proportion e_ne_simple if xrtreat==2
	estimates store app

	coefplot (dep, color(eltblue)) (stab, color(emidblue)) (app, color(edkblue)),  /// 
		vertical recast(bar) barwidth(0.25) finten(60) ///
		title("`name'", size(5))  ///
		yscale(range(0 0.65)) ylabel(0(0.1)0.6,labsize(4)) ytitle(Proportion, size(4)) ///
		xlabel(1 "Negative" 2 "Neither/Nor" 3 "Positive", angle(45) labsize(4)) /// 
		citop citype(logit) ciopt(lcolor(black) lwidth(vthin)) ///
		legend(label(1 "Depreciation") label(3 "No change") label(5 "Appreciation") size(4))
	graph save "econimp_`country'", replace	

	* - Personal situation
	
	proportion e_pe_simple if xrtreat==1
	estimates store dep
	proportion e_pe_simple if xrtreat==0
	estimates store stab
	proportion e_pe_simple if xrtreat==2
	estimates store app

	coefplot (dep, color(eltblue)) (stab, color(emidblue)) (app, color(edkblue)),  /// 
		vertical recast(bar) barwidth(0.25) finten(60) ///
		title("`name'", size(5))  ///
		yscale(range(0 0.65)) ylabel(0(0.1)0.6,labsize(4)) ytitle(Proportion, size(4)) ///
		xlabel(1 "Negative" 2 "Neither/Nor" 3 "Positive", angle(45) labsize(4)) /// 
		citop citype(logit) ciopt(lcolor(black) lwidth(vthin)) ///
		legend(label(1 "Depreciation") label(3 "No change") label(5 "Appreciation") size(4))
	graph save "persimp_`country'", replace
		
	restore
	
}

grc1leg govperf_MEX.gph govperf_IND.gph govperf_AUS.gph govperf_USA.gph govperf_JPN.gph , ///
	row(1)  ///
	pos(3) iscale(1.25) title("{bf: Government performance}", size(7)) ///
	graphregion(fcolor(white)) saving(govperf_ALL, replace)
graph display, xsize(13)	

grc1leg cbperf_MEX.gph cbperf_IND.gph cbperf_AUS.gph cbperf_USA.gph cbperf_JPN.gph , ///
	row(1)  ///
	pos(3) iscale(1.25) title("{bf: Central bank performance}", size(7)) ///
	graphregion(fcolor(white)) saving(cbperf_ALL, replace)
graph display, xsize(13)	

grc1leg vote_MEX.gph vote_IND.gph vote_AUS.gph vote_USA.gph vote_JPN.gph , ///
	row(1)  ///
	pos(3) iscale(1.25) title("{bf: Vote intentions}", size(7)) ///
	graphregion(fcolor(white)) saving(vote_ALL, replace)
graph display, xsize(13.0)	

grc1leg econimp_MEX.gph econimp_IND.gph econimp_AUS.gph econimp_USA.gph econimp_JPN.gph , ///
	row(1)  ///
	pos(3) iscale(1.25) title("{bf: National economy}", size(7)) ///
	graphregion(fcolor(white)) saving(econimp_ALL, replace)
graph display, xsize(13)	

grc1leg persimp_MEX.gph persimp_IND.gph persimp_AUS.gph persimp_USA.gph persimp_JPN.gph , ///
	row(1)  ///
	pos(3) iscale(1.25) title("{bf: Personal situation}", size(7)) ///
	graphregion(fcolor(white)) saving(persimp_ALL, replace)
graph display, xsize(13)	

grc1leg vote_ALL.gph ///
	govperf_ALL.gph ///
	cbperf_ALL.gph ///
	econimp_ALL.gph ///
	persimp_ALL.gph, ///
	row(5) ysize(10) scale(0.3) ///
	legendfrom(govperf_ALL.gph)
graph save FigureA20.gph, replace	

foreach file in cbperf_ALL.gph cbperf_AUS.gph cbperf_IND.gph cbperf_JPN.gph cbperf_MEX.gph cbperf_USA.gph ///
	econimp_ALL.gph econimp_AUS.gph econimp_IND.gph econimp_JPN.gph econimp_MEX.gph econimp_USA.gph ///
	govperf_ALL.gph govperf_AUS.gph govperf_IND.gph govperf_JPN.gph govperf_MEX.gph govperf_USA.gph ///
	persimp_ALL.gph persimp_AUS.gph persimp_IND.gph persimp_JPN.gph persimp_MEX.gph persimp_USA.gph ///
	vote_ALL.gph vote_AUS.gph vote_IND.gph vote_JPN.gph vote_MEX.gph vote_USA.gph {

	erase "`file'"
	
}	
	
drop e*_simple	

* Figure A21
* ----------
*
* - Government performance

reg e_pm ib0.xrtreat if country == "MEX"
reg e_pm ib0.xrtreat if country == "IND"
reg e_pm ib0.xrtreat if country == "AUS"
reg e_pm ib0.xrtreat if country == "USA"
reg e_pm ib0.xrtreat if country == "JPN"

* - CB performance

reg e_cb ib0.xrtreat if country == "MEX"
reg e_cb ib0.xrtreat if country == "IND"
reg e_cb ib0.xrtreat if country == "AUS"
reg e_cb ib0.xrtreat if country == "USA"
reg e_cb ib0.xrtreat if country == "JPN"

* - Plot results
*
* Note: see code for Figure 2 above for explanation why gsem command is used  

foreach var in e_pm e_cb {

	g `var'_AUS = `var' if country == "AUS"
	g `var'_IND = `var' if country == "IND"	
	g `var'_JPN = `var' if country == "JPN"
	g `var'_MEX = `var' if country == "MEX"
	g `var'_USA = `var' if country == "USA"

	gsem (`var'_MEX <- ib0.xrtreat) ///
		 (`var'_IND <- ib0.xrtreat) ///
		 (`var'_AUS <- ib0.xrtreat) ///
		 (`var'_USA <- ib0.xrtreat) ///
		 (`var'_JPN <- ib0.xrtreat)	 
		 
	estimates store `var'	
	
}	

coefplot e_pm, msymbol(O) omitted baselevel keep(*:) drop(_cons var*) ///
	coeflabels(0.xrtreat="No change" 1.xrtreat="Depreciation" 2.xrtreat="Appreciation", labsize(3)) ///
	eqlabels("{bf:Mexico}" "{bf:India}" "{bf:Australia}" "{bf:United States}" "{bf:Japan}", labsize(3.5) asheadings)	///
	mlabels(0.xrtreat= 1 "(ref. cat)") mlabel(@b) mlabpos(11) mlabgap(*1.5) format(%9.2f) ///
	xline(0, lcolor(black)) xscale(range(-0.6 0.6)) xlabel(-0.6(0.2)0.6, labsize(3) format(%9.1f)) ///
	title("{bf: Government performance}", size(3.5))  ///
	scheme(plotplain) saving(e_pm_plot.gph, replace) 
	
coefplot e_cb, msymbol(O) omitted baselevel keep(*:) drop(_cons var*) ///
	coeflabels(0.xrtreat="No change" 1.xrtreat="Depreciation" 2.xrtreat="Appreciation", labsize(3)) ///
	eqlabels("{bf:Mexico}" "{bf:India}" "{bf:Australia}" "{bf:United States}" "{bf:Japan}", labsize(3.5) asheadings)	///
	mlabel(@b) mlabpos(11) mlabgap(*1.5) format(%9.2f) ///
	xline(0, lcolor(black)) xscale(range(-0.6 0.4)) xlabel(-0.6(0.2)0.4, labsize(3) format(%9.1f)) ///
	title("{bf: CB performance}", size(3.5))  ///
	scheme(plotplain) saving(e_cb_plot.gph, replace) 
		
graph combine e_pm_plot.gph e_cb_plot.gph, ///
	row(1) imargin(medlarge)  ///
	graphregion(fcolor(white)) ///
	title("")	
graph display, xsize(6.25) scale(1.2) 
graph save FigureA21.gph, replace		

erase e_pm_plot.gph 
erase e_cb_plot.gph 

drop _est_e_pm _est_e_cb
drop e_pm_* e_cb_*


* Figure A22
* ----------
*
* - Tradeables sector, version 1

g trade_v1 = 0
replace trade_v1 = 1 if sector == 1 | sector == 3 | sector == 6 | sector == 8 | sector == 14 | sector == 16
replace trade_v1 = . if sector == .

* - Working population

g working = (employment == 1 | employment == 2)

* - Plot
*
* Note: only working population is considered

hist trade_v1 if country == "USA" & working == 1, discrete percent title({bf:USA}) /// 
	xtitle("Tradeables (v1 - tradeability)") xlabel(0 "no" 1 "yes", val) xsca(range(0 1.5)) ///
	xsize(5) saving(tradev1_usa.gph, replace) 

hist trade_v1 if country == "AUS" & working == 1, discrete percent title({bf:Australia}) /// 
	xtitle("Tradeables (v1 - tradeability)") xlabel(0 "no" 1 "yes", val) xsca(range(0 1.5)) ///
	xsize(5) saving(tradev1_aus.gph, replace) 

hist trade_v1 if country == "JPN" & working == 1, discrete percent title({bf:Japan}) /// 
	xtitle("Tradeables (v1 - tradeability)") xlabel(0 "no" 1 "yes", val) xsca(range(0 1.5)) ///
	xsize(5) saving(tradev1_jpn.gph, replace) 
	
hist trade_v1 if country == "MEX" & working == 1, discrete percent title({bf:Mexico}) /// 
	xtitle("Tradeables (v1 - tradeability)") xlabel(0 "no" 1 "yes", val) xsca(range(0 1.5)) ///
	xsize(5) saving(tradev1_mex.gph, replace) 	

hist trade_v1 if country == "IND" & working == 1, discrete percent title({bf:India}) /// 
	xtitle("Tradeables (v1 - tradeability)") xlabel(0 "no" 1 "yes", val) xsca(range(0 1.5)) ///
	xsize(5) saving(tradev1_ind.gph, replace) 	
	
graph combine tradev1_mex.gph tradev1_ind.gph tradev1_aus.gph tradev1_usa.gph tradev1_jpn.gph, ///
	col(2) holes(2) imargin(medium) iscale(0.5) ycommon ///
	graphregion(fcolor(white)) ///
	title("")	
graph display, ysize(8) scale(1) 	
graph save FigureA22.gph, replace	
	
erase tradev1_mex.gph
erase tradev1_ind.gph
erase tradev1_aus.gph
erase tradev1_usa.gph
erase tradev1_jpn.gph


* Figure A23
* ----------
*
* - Tradeables sector, version 2

g trade_aus = 0
replace trade_aus = 1 if sector == 1 | sector == 3 | sector == 4 | sector == 6 | sector == 8 | sector == 16
replace trade_aus = . if sector == .
replace trade_aus = . if country != "AUS"
tab sector trade_aus if country == "AUS", mis

g trade_jpn = 0
replace trade_jpn = 1 if sector == 4 | sector == 6 | sector == 8 | sector == 14 | sector == 16
replace trade_jpn = . if sector == .
replace trade_jpn = . if country != "JPN"
tab sector trade_jpn if country == "JPN", mis

g trade_usa = 0
replace trade_usa = 1 if sector == 1 | sector == 3 | sector == 4 | sector == 6 | sector == 8 | sector == 14 | sector == 16
replace trade_usa = . if sector == .
replace trade_usa = . if country != "USA"
tab sector trade_usa if country == "USA", mis

g trade_ind = 0
replace trade_ind = 1 if sector == 3 | sector == 6 | sector == 8 | sector == 14 | sector == 16
replace trade_ind = . if sector == .
replace trade_ind = . if country != "IND"
tab sector trade_ind if country == "IND", mis

g trade_mex = 0
replace trade_mex = 1 if sector == 1 | sector == 3 | sector == 8 
replace trade_mex = . if sector == .
replace trade_mex = . if country != "MEX"
tab sector trade_mex if country == "MEX", mis

g trade_v2 = .
replace trade_v2 = trade_aus if country == "AUS"
replace trade_v2 = trade_jpn if country == "JPN"
replace trade_v2 = trade_usa if country == "USA"
replace trade_v2 = trade_ind if country == "IND"
replace trade_v2 = trade_mex if country == "MEX"

drop trade_aus - trade_mex

* - Plot
*
* Note: only working population is considered

hist trade_v2 if country == "USA" & working == 1, discrete percent title({bf:USA}) /// 
	xtitle("Tradeables (v2 - RCA)") xlabel(0 "no" 1 "yes", val) xsca(range(0 1.5)) ///
	xsize(5) saving(tradev2_usa.gph, replace) 

hist trade_v2 if country == "AUS" & working == 1, discrete percent title({bf:Australia}) /// 
	xtitle("Tradeables (v2 - RCA)") xlabel(0 "no" 1 "yes", val) xsca(range(0 1.5)) ///
	xsize(5) saving(tradev2_aus.gph, replace) 

hist trade_v2 if country == "JPN" & working == 1, discrete  percent title({bf:Japan}) /// 
	xtitle("Tradeables (v2 - RCA)") xlabel(0 "no" 1 "yes", val) xsca(range(0 1.5)) ///
	xsize(5) saving(tradev2_jpn.gph, replace) 
	
hist trade_v2 if country == "MEX" & working == 1, discrete percent title({bf:Mexico}) /// 
	xtitle("Tradeables (v2 - RCA)") xlabel(0 "no" 1 "yes", val) xsca(range(0 1.5)) ///
	xsize(5) saving(tradev2_mex.gph, replace) 	

hist trade_v2 if country == "IND" & working == 1, discrete percent title({bf:India}) /// 
	xtitle("Tradeables (v2 - RCA)") xlabel(0 "no" 1 "yes", val) xsca(range(0 1.5)) ///
	xsize(5) saving(tradev2_ind.gph, replace) 	
	
graph combine tradev2_mex.gph tradev2_ind.gph tradev2_aus.gph tradev2_usa.gph tradev2_jpn.gph, ///
	col(2) holes(2) imargin(medium) iscale(0.5) ycommon ///
	graphregion(fcolor(white)) ///
	title("")	
graph display, ysize(8) scale(1) 
graph save FigureA23.gph, replace		
	
erase tradev2_mex.gph
erase tradev2_ind.gph
erase tradev2_aus.gph
erase tradev2_usa.gph
erase tradev2_jpn.gph


* Figure A24
* ----------
*
* - Controls: government support, gender, age

g govsupp_aus = .
g govsupp_usa = .
g govsupp_jpn = . 
g govsupp_ind = .
g govsupp_mex = .

tab vote if country == "AUS"
replace govsupp_aus = (vote == 2 | vote == 3) if country == "AUS" 
tab vote if country == "USA"
replace govsupp_usa = (vote == 1) if country == "USA"
tab vote if country == "JPN"
replace govsupp_jpn = (vote == 1 | vote == 3) if country == "JPN"
tab vote if country == "IND"
replace govsupp_ind = (vote == 3) if country == "IND"
tab vote if country == "MEX"
replace govsupp_mex = (vote == 103) if country == "MEX"

g female = (gender == 2)

local controls_mex "govsupp_mex age female"  
local controls_ind "govsupp_ind age female"
local controls_aus "govsupp_aus age female"
local controls_usa "govsupp_usa age female"
local controls_jpn "govsupp_jpn age female"

* - Plot results; 
*
* Notes: see code for Figure 2 above for explanation why gsem command is used  
* 		 only working population and those with a sectoral classification is considered

foreach var in e_vi e_ne e_pe {

	g `var'_MEX = `var' if country == "MEX" & trade_v1 == 1 & sector != 18 & working == 1
	g `var'_IND = `var' if country == "IND"	& trade_v1 == 1 & sector != 18 & working == 1
	g `var'_AUS = `var' if country == "AUS" & trade_v1 == 1 & sector != 18 & working == 1
	g `var'_USA = `var' if country == "USA" & trade_v1 == 1 & sector != 18 & working == 1
	g `var'_JPN = `var' if country == "JPN" & trade_v1 == 1 & sector != 18 & working == 1
	
	gsem 	(`var'_MEX <- ib0.xrtreat `controls_mex') ///
			(`var'_IND <- ib0.xrtreat `controls_ind') ///	
			(`var'_AUS <- ib0.xrtreat `controls_aus') ///
			(`var'_USA <- ib0.xrtreat `controls_usa') ///	
			(`var'_JPN <- ib0.xrtreat `controls_jpn') ///
							
	estimates store `var'_trade	
	
	drop `var'_AUS `var'_USA `var'_JPN `var'_IND `var'_MEX 
	
}	

foreach var in e_vi e_ne e_pe {

	g `var'_MEX = `var' if country == "MEX" & trade_v1 == 0 & sector != 18 & working == 1
	g `var'_IND = `var' if country == "IND"	& trade_v1 == 0 & sector != 18 & working == 1
	g `var'_AUS = `var' if country == "AUS" & trade_v1 == 0 & sector != 18 & working == 1
	g `var'_USA = `var' if country == "USA" & trade_v1 == 0 & sector != 18 & working == 1
	g `var'_JPN = `var' if country == "JPN" & trade_v1 == 0 & sector != 18 & working == 1

	gsem 	(`var'_MEX <- ib0.xrtreat `controls_mex') ///
			(`var'_IND <- ib0.xrtreat `controls_ind') ///		
			(`var'_AUS <- ib0.xrtreat `controls_aus') ///
			(`var'_USA <- ib0.xrtreat `controls_usa') ///	
			(`var'_JPN <- ib0.xrtreat `controls_jpn')
					
			
					
	estimates store `var'_notrade	
	
	drop `var'_AUS `var'_USA `var'_JPN `var'_IND `var'_MEX 
	
}	

coefplot (e_vi_trade, label("Tradeables") msymbol(O) mcolor(black)) (e_vi_notrade, label("Non-tradeables") msymbol(D)),  ///
	omitted baselevel keep(*:0.xrtreat *:1.xrtreat *:2.xrtreat )  ///
	coeflabels(0.xrtreat="No change" 1.xrtreat="Depreciation" 2.xrtreat="Appreciation", labsize(3)) ///
	eqlabels("{bf:Mexico}" "{bf:India}" "{bf:Australia}" "{bf:United States}" "{bf:Japan}", labsize(3.5) asheadings)	///
	xline(0, lcolor(black)) xlabel(, labsize(3) format(%9.1f)) ///
	title("{bf: Vote intentions}", size(3.5)) ///	
	xsize(5) ///	
	scheme(plotplain) saving(vi_bytrade_v1, replace)
	
coefplot (e_ne_trade, label("Tradeables") msymbol(O) mcolor(black)) (e_ne_notrade, label("Non-tradeables") msymbol(D)),  ///
	omitted baselevel keep(*:0.xrtreat *:1.xrtreat *:2.xrtreat )  ///
	coeflabels(0.xrtreat="No change" 1.xrtreat="Depreciation" 2.xrtreat="Appreciation", labsize(3)) ///
	eqlabels("{bf:Mexico}" "{bf:India}" "{bf:Australia}" "{bf:United States}" "{bf:Japan}", labsize(3.5) asheadings)	///
	xline(0, lcolor(black)) xlabel(, labsize(3) format(%9.1f)) ///
	title("{bf: National economy}", size(3.5)) ///	
	xsize(5) ///	
	scheme(plotplain) saving(ne_bytrade_v1, replace)	
	
coefplot (e_pe_trade, label("Tradeables") msymbol(O) mcolor(black)) (e_pe_notrade, label("Non-tradeables") msymbol(D)),  ///
	omitted baselevel keep(*:0.xrtreat *:1.xrtreat *:2.xrtreat )  ///
	coeflabels(0.xrtreat="No change" 1.xrtreat="Depreciation" 2.xrtreat="Appreciation", labsize(3)) ///
	eqlabels("{bf:Mexico}" "{bf:India}" "{bf:Australia}" "{bf:United States}" "{bf:Japan}", labsize(3.5) asheadings)	///
	xline(0, lcolor(black)) xlabel(, labsize(3) format(%9.1f)) ///
	title("{bf: Personal situation}", size(3.5)) ///	
	xsize(5) ///	
	scheme(plotplain) saving(ps_bytrade_v1, replace)	

grc1leg vi_bytrade_v1.gph ne_bytrade_v1.gph ps_bytrade_v1.gph, ///
	row(1) imargin(medlarge)  ///
	graphregion(fcolor(white)) ///
	title("")	
graph display, xsize(9.25) scale(1.2)
graph save FigureA24.gph, replace	 	
				
erase vi_bytrade_v1.gph
erase ne_bytrade_v1.gph
erase ps_bytrade_v1.gph	

drop _est_e_*_trade _est_e_*_notrade	


* Figure A25
* ----------
*
* - Controls: as above

local controls_mex "govsupp_mex age female"  
local controls_ind "govsupp_ind age female"
local controls_aus "govsupp_aus age female"
local controls_usa "govsupp_usa age female"
local controls_jpn "govsupp_jpn age female"

* - Plot results; 
*
* Notes: see code for Figure 2 above for explanation why gsem command is used  
* 		 only working population and those with a sectoral classification is considered

foreach var in e_vi e_ne e_pe {

	g `var'_MEX = `var' if country == "MEX" & trade_v2 == 1 & sector != 18 & working == 1
	g `var'_IND = `var' if country == "IND" & trade_v2 == 1 & sector != 18 & working == 1
	g `var'_AUS = `var' if country == "AUS" & trade_v2 == 1 & sector != 18 & working == 1
	g `var'_USA = `var' if country == "USA" & trade_v2 == 1 & sector != 18 & working == 1
	g `var'_JPN = `var' if country == "JPN" & trade_v2 == 1 & sector != 18 & working == 1
	
	gsem 	(`var'_MEX <- ib0.xrtreat `controls_mex') ///
			(`var'_IND <- ib0.xrtreat `controls_ind') ///	
			(`var'_AUS <- ib0.xrtreat `controls_aus') ///
			(`var'_USA <- ib0.xrtreat `controls_usa') ///	
			(`var'_JPN <- ib0.xrtreat `controls_jpn') ///
						
	estimates store `var'_trade	
	
	drop `var'_AUS `var'_USA `var'_JPN `var'_IND `var'_MEX 
	
}	

foreach var in e_pm e_cb e_vi e_ne e_pe {

	g `var'_MEX = `var' if country == "MEX" & trade_v2 == 0 & sector != 18 & working == 1
	g `var'_IND = `var' if country == "IND" & trade_v2 == 0 & sector != 18 & working == 1
	g `var'_AUS = `var' if country == "AUS" & trade_v2 == 0 & sector != 18 & working == 1
	g `var'_USA = `var' if country == "USA" & trade_v2 == 0 & sector != 18 & working == 1
	g `var'_JPN = `var' if country == "JPN" & trade_v2 == 0 & sector != 18 & working == 1

	gsem 	(`var'_MEX <- ib0.xrtreat `controls_mex') ///
			(`var'_IND <- ib0.xrtreat `controls_ind') ///		
			(`var'_AUS <- ib0.xrtreat `controls_aus') ///
			(`var'_USA <- ib0.xrtreat `controls_usa') ///	
			(`var'_JPN <- ib0.xrtreat `controls_jpn')
									
	estimates store `var'_notrade	
	
	drop `var'_AUS `var'_USA `var'_JPN `var'_IND `var'_MEX 
	
}	

coefplot (e_vi_trade, label("Tradeables (revised)") msymbol(O) mcolor(black)) (e_vi_notrade, label("Non-tradeables") msymbol(D)),  ///
	omitted baselevel keep(*:0.xrtreat *:1.xrtreat *:2.xrtreat )  ///
	coeflabels(0.xrtreat="No change" 1.xrtreat="Depreciation" 2.xrtreat="Appreciation", labsize(3)) ///
	eqlabels("{bf:Mexico}" "{bf:India}" "{bf:Australia}" "{bf:United States}" "{bf:Japan}", labsize(3.5) asheadings)	///
	xline(0, lcolor(black)) xlabel(-0.5 0 0.5, labsize(3) format(%9.1f)) ///
	title("{bf: Vote intentions}", size(3.5)) ///	
	xsize(5) ///	
	scheme(plotplain) saving(vi_bytrade_v2, replace)
	
coefplot (e_ne_trade, label("Tradeables (revised)") msymbol(O) mcolor(black)) (e_ne_notrade, label("Non-tradeables") msymbol(D)),  ///
	omitted baselevel keep(*:0.xrtreat *:1.xrtreat *:2.xrtreat )  ///
	coeflabels(0.xrtreat="No change" 1.xrtreat="Depreciation" 2.xrtreat="Appreciation", labsize(3)) ///
	eqlabels("{bf:Mexico}" "{bf:India}" "{bf:Australia}" "{bf:United States}" "{bf:Japan}", labsize(3.5) asheadings)	///
	xline(0, lcolor(black)) xlabel(-0.5 0 0.5 1, labsize(3) format(%9.1f)) ///
	title("{bf: National economy}", size(3.5)) ///	
	xsize(5) ///	
	scheme(plotplain) saving(ne_bytrade_v2, replace)
	
coefplot (e_pe_trade, label("Tradeables (revised)") msymbol(O) mcolor(black)) (e_pe_notrade, label("Non-tradeables") msymbol(D)),  ///
	omitted baselevel keep(*:0.xrtreat *:1.xrtreat *:2.xrtreat )  ///
	coeflabels(0.xrtreat="No change" 1.xrtreat="Depreciation" 2.xrtreat="Appreciation", labsize(3)) ///
	eqlabels("{bf:Mexico}" "{bf:India}" "{bf:Australia}" "{bf:United States}" "{bf:Japan}", labsize(3.5) asheadings)	///
	xline(0, lcolor(black)) xlabel(-0.5 0 0.5 1, labsize(3) format(%9.1f)) ///
	title("{bf: Personal situation}", size(3.5)) ///	
	xsize(5) ///	
	scheme(plotplain) saving(ps_bytrade_v2, replace)		

grc1leg vi_bytrade_v2.gph ne_bytrade_v2.gph ps_bytrade_v2.gph, ///
	row(1) imargin(medlarge)  ///
	graphregion(fcolor(white)) ///
	title("")	
graph display, xsize(9.25) scale(1.2) 	
graph save FigureA25.gph, replace	
				
erase vi_bytrade_v2.gph
erase ne_bytrade_v2.gph
erase ps_bytrade_v2.gph	
	
drop _est_e_*_trade _est_e_*_notrade	


* Figure A26
* ----------
*
* - High education variable (= at least full college)

g highedu = 0 if education <= 4 & education != .
replace highedu = 1 if education > 4 & education != .

* - Controls: as above

local controls_mex "govsupp_mex age female"  
local controls_ind "govsupp_ind age female"
local controls_aus "govsupp_aus age female"
local controls_usa "govsupp_usa age female"
local controls_jpn "govsupp_jpn age female"

* - Plot results; 
*
* Notes: see code for Figure 2 above for explanation why gsem command is used  

foreach var in e_vi e_ne e_pe {

	g `var'_MEX = `var' if country == "MEX" & highedu == 1
	g `var'_IND = `var' if country == "IND"	& highedu == 1
	g `var'_AUS = `var' if country == "AUS" & highedu == 1
	g `var'_USA = `var' if country == "USA" & highedu == 1
	g `var'_JPN = `var' if country == "JPN" & highedu == 1
	
	gsem 	(`var'_MEX <- ib0.xrtreat `controls_mex') ///
			(`var'_IND <- ib0.xrtreat `controls_ind') ///	
			(`var'_AUS <- ib0.xrtreat `controls_aus') ///
			(`var'_USA <- ib0.xrtreat `controls_usa') ///	
			(`var'_JPN <- ib0.xrtreat `controls_jpn') ///
						
	estimates store `var'_highedu
	
	drop `var'_AUS `var'_USA `var'_JPN `var'_IND `var'_MEX 
	
}	

foreach var in e_pm e_cb e_vi e_ne e_pe {

	g `var'_MEX = `var' if country == "MEX" & highedu == 0
	g `var'_IND = `var' if country == "IND"	& highedu == 0
	g `var'_AUS = `var' if country == "AUS" & highedu == 0
	g `var'_USA = `var' if country == "USA" & highedu == 0
	g `var'_JPN = `var' if country == "JPN" & highedu == 0

	gsem 	(`var'_MEX <- ib0.xrtreat `controls_mex') ///
			(`var'_IND <- ib0.xrtreat `controls_ind') ///		
			(`var'_AUS <- ib0.xrtreat `controls_aus') ///
			(`var'_USA <- ib0.xrtreat `controls_usa') ///	
			(`var'_JPN <- ib0.xrtreat `controls_jpn')
									
	estimates store `var'_lowedu	
	
	drop `var'_AUS `var'_USA `var'_JPN `var'_IND `var'_MEX 
	
}	

coefplot (e_vi_highedu, label("High education") msymbol(O) mcolor(black)) (e_vi_lowedu, label("Low education") msymbol(D)),  ///
	omitted baselevel keep(*:0.xrtreat *:1.xrtreat *:2.xrtreat )  ///
	coeflabels(0.xrtreat="No change" 1.xrtreat="Depreciation" 2.xrtreat="Appreciation", labsize(3)) ///
	eqlabels("{bf:Mexico}" "{bf:India}" "{bf:Australia}" "{bf:United States}" "{bf:Japan}", labsize(3.5) asheadings)	///
	xline(0, lcolor(black)) xlabel(, labsize(3) format(%9.1f)) ///
	title("{bf: Vote intentions}", size(3.5)) ///	
	xsize(5) ///	
	scheme(plotplain) saving(vi_byeducation, replace)	
	
coefplot (e_ne_highedu, label("High education") msymbol(O) mcolor(black)) (e_ne_lowedu, label("Low education") msymbol(D)),  ///
	omitted baselevel keep(*:0.xrtreat *:1.xrtreat *:2.xrtreat )  ///
	coeflabels(0.xrtreat="No change" 1.xrtreat="Depreciation" 2.xrtreat="Appreciation", labsize(3)) ///
	eqlabels("{bf:Mexico}" "{bf:India}" "{bf:Australia}" "{bf:United States}" "{bf:Japan}", labsize(3.5) asheadings)	///
	xline(0, lcolor(black)) xlabel(, labsize(3) format(%9.1f)) ///
	title("{bf: National economy}", size(3.5)) ///	
	xsize(5) ///	
	scheme(plotplain) saving(ne_byeducation, replace)	
	
coefplot (e_pe_highedu, label("High education") msymbol(O) mcolor(black)) (e_pe_lowedu, label("Low education") msymbol(D)),  ///
	omitted baselevel keep(*:0.xrtreat *:1.xrtreat *:2.xrtreat )  ///
	coeflabels(0.xrtreat="No change" 1.xrtreat="Depreciation" 2.xrtreat="Appreciation", labsize(3)) ///
	eqlabels("{bf:Mexico}" "{bf:India}" "{bf:Australia}" "{bf:United States}" "{bf:Japan}", labsize(3.5) asheadings)	///
	xline(0, lcolor(black)) xlabel(, labsize(3) format(%9.1f)) ///
	title("{bf: Personal situation}", size(3.5)) ///	
	xsize(5) ///	
	scheme(plotplain) saving(ps_byeducation, replace)		

grc1leg vi_byeducation.gph ne_byeducation.gph ps_byeducation.gph, ///
	row(1) imargin(medlarge)  ///
	graphregion(fcolor(white)) ///
	title("")	
graph display, xsize(9.25) scale(1.2) 
graph save FigureA26.gph, replace		
				
erase vi_byeducation.gph
erase ne_byeducation.gph
erase ps_byeducation.gph	

drop _est_e_*_lowedu _est_e_*_highedu
	

* Figure A27
* ----------	

hist attention if country == "USA", title({bf:USA}) discrete percent /// 
	xtitle("Comprehension") xlabel(0 1, val) xsca(range(0.5 1.5)) ///
	xsize(5) saving(attention_usa.gph, replace) 
	
hist attention if country == "AUS", title({bf:Australia}) discrete percent /// 
	xtitle("Comprehension") xlabel(0 1, val) xsca(range(0.5 1.5)) ///
	xsize(5) saving(attention_aus.gph, replace) 
	
hist attention if country == "IND", title({bf:India}) discrete percent /// 
	xtitle("Comprehension") xlabel(0 1, val) xsca(range(0.5 1.5)) ///
	xsize(5) saving(attention_ind.gph, replace) 
	
hist attention if country == "MEX", title({bf:Mexico}) discrete percent /// 
	xtitle("Comprehension") xlabel(0 1, val) xsca(range(0.5 1.5)) ///
	xsize(5) saving(attention_mex.gph, replace) 
	
hist attention if country == "JPN", title({bf:Japan}) discrete percent /// 
	xtitle("Comprehension") xlabel(0 1, val) xsca(range(0.5 1.5)) ///
	xsize(5) saving(attention_jpn.gph, replace) 

graph combine attention_mex.gph attention_ind.gph attention_aus.gph attention_usa.gph attention_jpn.gph, ///
	col(2) holes(2) imargin(medium) iscale(0.5) ycommon ///
	graphregion(fcolor(white)) ///
	title("")	
graph display, ysize(8) scale(1) scheme(plotplain)	
graph save FigureA27.gph, replace	

erase attention_mex.gph
erase attention_ind.gph
erase attention_aus.gph
erase attention_usa.gph
erase attention_jpn.gph

	
* Figure A28
* ----------
*
* Note: see code for Figure 2 above for explanation why gsem command is used  
*
* - (a) Attention check passed

g e_vi_AUS = e_vi if country == "AUS" & attention == 1
g e_vi_IND = e_vi if country == "IND" & attention == 1	
g e_vi_JPN = e_vi if country == "JPN" & attention == 1
g e_vi_MEX = e_vi if country == "MEX" & attention == 1
g e_vi_USA = e_vi if country == "USA" & attention == 1

gsem	(e_vi_MEX <- ib0.xrtreat) ///
		(e_vi_IND <- ib0.xrtreat) ///
		(e_vi_AUS <- ib0.xrtreat) ///
		(e_vi_USA <- ib0.xrtreat) ///
		(e_vi_JPN <- ib0.xrtreat)	 
		 
estimates store e_vi_attyes	
	
drop e_vi_AUS e_vi_USA e_vi_JPN e_vi_IND e_vi_MEX 
	
coefplot e_vi_attyes, msymbol(O) omitted baselevel keep(*:) drop(_cons var*) ///
	coeflabels(0.xrtreat="No change" 1.xrtreat="Depreciation" 2.xrtreat="Appreciation", labsize(3)) ///
	eqlabels("{bf:Mexico}" "{bf:India}" "{bf:Australia}" "{bf:United States}" "{bf:Japan}", labsize(3.5) asheadings)	///
	mlabel mlabpos(11) mlabgap(*1.5) format(%9.2f) ///
	xline(0, lcolor(black)) xscale(range(-0.3 0.3)) xlabel(-0.3(0.1)0.3, labsize(3) format(%9.1f)) ///
	title("{bf: Vote intentions}", size(3.5)) ///
	xsize(5) ///
	scheme(plotplain) 	
graph save FigureA28a.gph, replace		
		
* - (b) Attention check failed

g e_vi_AUS = e_vi if country == "AUS" & attention == 0
g e_vi_IND = e_vi if country == "IND" & attention == 0	
g e_vi_JPN = e_vi if country == "JPN" & attention == 0
g e_vi_MEX = e_vi if country == "MEX" & attention == 0
g e_vi_USA = e_vi if country == "USA" & attention == 0

gsem	(e_vi_MEX <- ib0.xrtreat) ///
		(e_vi_IND <- ib0.xrtreat) ///
		(e_vi_AUS <- ib0.xrtreat) ///
		(e_vi_USA <- ib0.xrtreat) ///
		(e_vi_JPN <- ib0.xrtreat)	 
		 
estimates store e_vi_attno	
	
drop e_vi_AUS e_vi_USA e_vi_JPN e_vi_IND e_vi_MEX 
	
coefplot e_vi_attno, msymbol(O) omitted baselevel keep(*:) drop(_cons var*) ///
	coeflabels(0.xrtreat="No change" 1.xrtreat="Depreciation" 2.xrtreat="Appreciation", labsize(3)) ///
	eqlabels("{bf:Mexico}" "{bf:India}" "{bf:Australia}" "{bf:United States}" "{bf:Japan}", labsize(3.5) asheadings)	///
	mlabel mlabpos(11) mlabgap(*1.5) format(%9.2f) ///
	xline(0, lcolor(black)) xscale(range(-0.3 0.3)) xlabel(-0.3(0.1)0.3, labsize(3) format(%9.1f)) ///
	title("{bf: Vote intentions}", size(3.5)) ///
	xsize(5) ///
	scheme(plotplain) 	
graph save FigureA28b.gph, replace		
	
drop _est_e_vi_attyes _est_e_vi_attno	

	
* Figure A29
* ----------
*
* Notes: see code for Figure 2 above for explanation why gsem command is used  
*
* - (a) Standard

g e_vi_AUS = e_vi if country == "AUS" & reverse == 0
g e_vi_IND = e_vi if country == "IND" & reverse == 0	
g e_vi_JPN = e_vi if country == "JPN" & reverse == 0
g e_vi_MEX = e_vi if country == "MEX" & reverse == 0
g e_vi_USA = e_vi if country == "USA" & reverse == 0

gsem	(e_vi_MEX <- ib0.xrtreat) ///
		(e_vi_IND <- ib0.xrtreat) ///
		(e_vi_AUS <- ib0.xrtreat) ///
		(e_vi_USA <- ib0.xrtreat) ///
		(e_vi_JPN <- ib0.xrtreat)	 
		 
estimates store e_vi_revno	
	
drop e_vi_AUS e_vi_USA e_vi_JPN e_vi_IND e_vi_MEX 
	
coefplot e_vi_revno, msymbol(O) omitted baselevel keep(*:) drop(_cons var*) ///
	coeflabels(0.xrtreat="No change" 1.xrtreat="Depreciation" 2.xrtreat="Appreciation", labsize(3)) ///
	eqlabels("{bf:Mexico}" "{bf:India}" "{bf:Australia}" "{bf:United States}" "{bf:Japan}", labsize(3.5) asheadings)	///
	mlabel mlabpos(11) mlabgap(*1.5) format(%9.2f) ///
	xline(0, lcolor(black)) xscale(range(-0.3 0.3)) xlabel(-0.3(0.1)0.3, labsize(3) format(%9.1f)) ///
	title("{bf: Vote intentions}", size(3.5)) ///
	xsize(5) ///
	scheme(plotplain) 	
graph save FigureA29a.gph, replace		
		
* - (b) Reverse

g e_vi_AUS = e_vi if country == "AUS" & reverse == 1
g e_vi_IND = e_vi if country == "IND" & reverse == 1
g e_vi_JPN = e_vi if country == "JPN" & reverse == 1
g e_vi_MEX = e_vi if country == "MEX" & reverse == 1
g e_vi_USA = e_vi if country == "USA" & reverse == 1

gsem	(e_vi_MEX <- ib0.xrtreat) ///
		(e_vi_IND <- ib0.xrtreat) ///
		(e_vi_AUS <- ib0.xrtreat) ///
		(e_vi_USA <- ib0.xrtreat) ///
		(e_vi_JPN <- ib0.xrtreat)	 
		 
estimates store e_vi_revyes	
	
drop e_vi_AUS e_vi_USA e_vi_JPN e_vi_IND e_vi_MEX 
	
coefplot e_vi_revyes, msymbol(O) omitted baselevel keep(*:) drop(_cons var*) ///
	coeflabels(0.xrtreat="No change" 1.xrtreat="Depreciation" 2.xrtreat="Appreciation", labsize(3)) ///
	eqlabels("{bf:Mexico}" "{bf:India}" "{bf:Australia}" "{bf:United States}" "{bf:Japan}", labsize(3.5) asheadings)	///
	mlabel mlabpos(11) mlabgap(*1.5) format(%9.2f) ///
	xline(0, lcolor(black)) xscale(range(-0.3 0.3)) xlabel(-0.3(0.1)0.3, labsize(3) format(%9.1f)) ///
	title("{bf: Vote intentions}", size(3.5)) ///
	xsize(5) ///
	scheme(plotplain) 
graph save FigureA29b.gph, replace		

drop _est_e_vi_revno _est_e_vi_revyes




